
	Reason for change:
	In the current TTCN implementation of 8.1.5.4.1
SS transmits 10 IP packets on each DRB (using QFI2 for DRB1 and QFI5 for DRB2) . 
At step2 ,SS may receive the set of loop back packets in the same slot or in multiple slots  

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc79577525]Corrections required
0.1 [bookmark: _Toc79577526]Correction to module RRC_Others_NR5GC
	Module name
	RRC_Others_NR5GC

	Reason for change
	NR_DRB_COMMON_IND data type has to be used for the handling of multiple IP packets in the same slot or in multiple slots .Importing NR_ASP_DrbDefs is required for the compilation purpose


	Summary of change
	Import of NR_ASP_DrbDefs is added in the module RRC_Others_NR5GC

	TTCN module
	NR5GC\8_1_5\RRC_Others_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	  import from NG_NAS_MsgContainers all;
  import from NAS_CommonTypeDefs all;
  import from NAS_CommonTemplates all;
  import from NR_SecurityFunctions all;
  import from EUTRA_NR_Parameters all;
  import from NR_ASP_Templates_CommonPart all;
  import from NR_DCI_Configuration all;



After Change:
	  import from NG_NAS_MsgContainers all;
  import from NAS_CommonTypeDefs all;
  import from NAS_CommonTemplates all;
  import from NR_SecurityFunctions all;
  import from EUTRA_NR_Parameters all;
  import from NR_ASP_Templates_CommonPart all;
  import from NR_DCI_Configuration all;
  import from NR_ASP_DrbDefs all;


0.2 [bookmark: _Toc79577527]Correction to function f_TC_8_1_5_4_1_NR5GC_TestBody
	Function name
	f_TC_8_1_5_4_1_NR5GC_TestBody

	Reason for change
	In the current TTCN implementation of 8.1.5.4.1
The SS transmits 10 IP packets on each DRB (using QFI2 for DRB1 and QFI5 for DRB2) . 
At step2 ,SS may receive the set of loop back packets in the same slot or in multiple slots  

	Summary of change
	TTCN is modified to handle set of loop back packets in the same slot or in multiple slots ;Also added pass verdicts based on the number of packets received associated to QFI2 and QFI5

	TTCN module
	Ttcn\develop\NR5GC\8_1_5\RRC_Others_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	function f_TC_8_1_5_4_1_NR5GC_TestBody (IntegerList_Type p_DRBList, integer p_PduSessionId) runs on NR_BASE_PTC
  {
   var template (value) DRB_CountMSB_InfoList v_CountList;
   var integer v_FirstDRB := p_DRBList[0];
   var integer v_SecondDRB := p_DRBList[1];
   var integer v_ThirdDRB := p_DRBList[2]; // To be used as not established DRB
   // @sic R5s200630 sic@
   var octetstring v_IPDataQFI2 := f_SdapTests_CreateDatagram("3"); // Packet with remote address matching filter #3 @sic R5s200733 sic@
   var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4
   var template (value) QosFlow_Identification_Type v_QoS_Id2 := cs_IP_QosFlowId(p_PduSessionId, 2); // QFI2 is mapped to First DRB @sic R5s200733 sic@
   var template (value) QosFlow_Identification_Type v_QoS_Id5 := cs_IP_QosFlowId(p_PduSessionId, 5); // QFI5 is mapped to Second DRB
   var integer i;
   var integer v_IPDataCount := 0;
   
   // @siclog "Step 1" siclog@
   // The SS transmits 10 IP packets on each DRB (using QF1 for DRB1 and QFI5 for DRB2)
   for (i:=0; i < 10; i:=i+1) {
     DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id2, {v_IPDataQFI2})); // @sic R5s200630, R5s200733 sic@
   }
   for (i:=0; i < 10; i:=i+1) {
     DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5})); // @sic R5s200630 sic@
   }
  
   // @siclog "Step 2" siclog@
   // The UE loops back the IP packets received in step 1
   do {
     DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, (v_QoS_Id2, v_QoS_Id5), ({v_IPDataQFI2}, {v_IPDataQFI5}))); // @sic R5s200630, R5s200733, R5s200978 sic@
     v_IPDataCount := v_IPDataCount + 1;
   } while (v_IPDataCount < 20);
   
   // @siclog "Step 3" siclog@
   // The SS transmits CounterCheck message with drb-Identity not included in the drb-CountMSB-InfoList
   v_CountList := {cs_38508_DRB_CountMSB_Info(v_FirstDRB, 1, 1),
                   cs_38508_DRB_CountMSB_Info(v_ThirdDRB, 3, 3)};
                  
   SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_CounterCheck(tsc_NR_RRC_TI_Def, v_CountList)));
  
   // @siclog "Step 4" siclog@
   // Check: Does the UE send CounterCheckResponse message by including the drb-Identity, the count-Uplink and the count-Downlink
   // set to the value of TX_NEXT and RX_NEXT" 1 in drb-CountInfoList IE for each DRB established?
   SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_CounterCheckResponse(tsc_NR_RRC_TI_Def,
                                                                              {permutation (cr_DRB_CountInfo(v_FirstDRB, 9, 9), //@sic R5s200978 sic@
                                                                               cr_DRB_CountInfo(v_SecondDRB, 9, 9), //@sic R5s200978 sic@
                                                                               cr_DRB_CountInfo(v_ThirdDRB, 384, 384))}))); //@sic R5s200733 sic@
   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3");
  }// end of f_TC_8_1_5_4_1_NR5GC_TestBody



After Change:
	function f_TC_8_1_5_4_1_NR5GC_TestBody (IntegerList_Type p_DRBList, integer p_PduSessionId) runs on NR_BASE_PTC
  {
   var template (value) DRB_CountMSB_InfoList v_CountList;
   var integer v_FirstDRB := p_DRBList[0];
   var integer v_SecondDRB := p_DRBList[1];
   var integer v_ThirdDRB := p_DRBList[2]; // To be used as not established DRB
   // @sic R5s200630 sic@
   var octetstring v_IPDataQFI2 := f_SdapTests_CreateDatagram("3"); // Packet with remote address matching filter #3 @sic R5s200733 sic@
   var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4
   var template (value) QosFlow_Identification_Type v_QoS_Id2 := cs_IP_QosFlowId(p_PduSessionId, 2); // QFI2 is mapped to First DRB @sic R5s200733 sic@
   var template (value) QosFlow_Identification_Type v_QoS_Id5 := cs_IP_QosFlowId(p_PduSessionId, 5); // QFI5 is mapped to Second DRB
   var integer i;
   var integer v_IPDataCount := 0;
   var integer v_IPDataCount_QFI2 := 0;
   var integer v_IPDataCount_QFI5 := 0;
   var NR_DRB_COMMON_IND v_ReceivedPdu1;
   var integer received_packets := 0;
   var octetstring v_IPDataQFIx := '00'O;

   // @siclog "Step 1" siclog@
   // The SS transmits 10 IP packets on each DRB (using QF1 for DRB1 and QFI5 for DRB2)
   for (i:=0; i < 10; i:=i+1) {
     DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id2, {v_IPDataQFI2})); // @sic R5s200630, R5s200733 sic@
   }
   for (i:=0; i < 10; i:=i+1) {
     DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id5, {v_IPDataQFI5})); // @sic R5s200630 sic@
   }
  
   // @siclog "Step 2" siclog@
   // The UE loops back the IP packets received in step 1
   do {
     DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, (v_QoS_Id2, v_QoS_Id5), ?)) -> value v_ReceivedPdu1 ;// @sic R5s200630, R5s200733, R5s200978 sic@
     {
         received_packets := lengthof(v_ReceivedPdu1.U_Plane.SlotData.PduSduList.SdapSdu);
         for ( i:=0 ; i < received_packets ; i := i+1 )
         {
             v_IPDataQFIx := v_ReceivedPdu1.U_Plane.SlotData.PduSduList.SdapSdu[i];
             if ((v_IPDataQFIx == v_IPDataQFI2) and (v_IPDataCount_QFI2 < 10)) 
             {
               v_IPDataCount_QFI2 := v_IPDataCount_QFI2 + 1 ;
               if (v_IPDataCount_QFI2 == 10)
               {
                   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2 - SS Received all 10 IP packets DRB using QF2 ");  
               }
             }
             else if ((v_IPDataQFIx == v_IPDataQFI5) and (v_IPDataCount_QFI5 < 10))
             {
               v_IPDataCount_QFI5 := v_IPDataCount_QFI5 + 1 ;
               if (v_IPDataCount_QFI5 == 10)
               {
                   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2 - SS Received all 10 IP packets using QF5 ");  
               }
             }
             else
             {
                 f_SetVerdict(fail,__FILE__,__LINE__,"Unexpected Packet is received at SS");
             }
         }
     }
     v_IPDataCount := v_IPDataCount + received_packets;
   }while (v_IPDataCount < 20);
   
   // @siclog "Step 3" siclog@
   // The SS transmits CounterCheck message with drb-Identity not included in the drb-CountMSB-InfoList
   v_CountList := {cs_38508_DRB_CountMSB_Info(v_FirstDRB, 1, 1),
                   cs_38508_DRB_CountMSB_Info(v_ThirdDRB, 3, 3)};
                  
   SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_CounterCheck(tsc_NR_RRC_TI_Def, v_CountList)));
  
   // @siclog "Step 4" siclog@
   // Check: Does the UE send CounterCheckResponse message by including the drb-Identity, the count-Uplink and the count-Downlink
   // set to the value of TX_NEXT and RX_NEXT" 1 in drb-CountInfoList IE for each DRB established?
   SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_CounterCheckResponse(tsc_NR_RRC_TI_Def,
                                                                              {permutation (cr_DRB_CountInfo(v_FirstDRB, 9, 9), //@sic R5s200978 sic@
                                                                               cr_DRB_CountInfo(v_SecondDRB, 9, 9), //@sic R5s200978 sic@
                                                                               cr_DRB_CountInfo(v_ThirdDRB, 384, 384))}))); //@sic R5s200733 sic@
   f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3");
  }// end of f_TC_8_1_5_4_1_NR5GC_TestBody



